Examples for Blast 2022 Talk

In[∘]:=
```
Needs["conglattice`"]; Needs["mma2uacalc`"];
```

Identifying semilattices and the semilattice replica

Given a binar, is it a semilattice?

In[∘]:=
```
IdempotentQ[t_] := Module[{n = Length[t]},
  AllTrue[Range[n], (t[[#, #]] == #) &]]
```

In[∘]:=
```
CommutativeQ[t_] := (t == Transpose[t])
```

In[∘]:=
```
(* assoctest[t_][{x_,y_,z_}]:=(t[[x,t[[y,z]]]]==t[[t[[x,y]],z]])
AssociativeQ[t_]:=Module[{n=Length[t]},
   AllTrue[Tuples[Range[n],3],assoctest[t]]]*)
```

In[∘]:=
```
AssociativeQ[t_] := With[{n = Length[t]}, Module[{flag = True},
   Do[If[t[[x, t[[y, z]]]] ≠ t[[t[[x, y]], z]], flag = False; Break[] ],
    {x, 1, n}, {y, 1, n}, {z, 1, n}];
   flag]]
```

In[∘]:=
```
SemilatticeQ[t_] := IdempotentQ[t] && CommutativeQ[t] && AssociativeQ[t]
```

Compute the S-verbal congruence on a binar

In[∘]:=
```
slpairs[t_] := Module[{n = Length[t], x, y, z},
  DeleteDuplicates[Join[Table[{t[[x, x]], x}, {x, 1, n}],
    Flatten[Table[{t[[x, y]], t[[y, x]]}, {x, 1, n}, {y, 1, n}], 1],
    Flatten[Table[
      {t[[t[[x, y]], z]], t[[x, t[[y, z]]]]}, {x, 1, n}, {y, 1, n}, {z, 1, n}], 2]]]]
```

In[∘]:=
```
Sverbal[t_] := cg[slpairs[t], ETlist[t], Length[t]]
```

Unordered pairs from a congruence

In[∘]:=
```
pairs[V_forest] :=
 Flatten[DeleteCases[List @@ (Subsets[#, {2}] & /@ forest2part[V]), {}], 1]
```

Given a subset (hopefully a subbinar), return a new subtable

Probably, s needs to be sorted in ascending order for this to work

```
In[]:= subtab[t_, s_] := Module[{n = Length[s]},
         t[[s, s]] /. Thread[s → Range[n]]]
```

Testing for membership in S*S and in (S*S)*S

```
In[]:= SSQ[t_] := Module[{rho},
         rho = forest2part[Sverbal[t]];
         AllTrue[rho, SemilatticeQ[subtab[t, #]] &]]

In[]:= SS3Q[t_] := Module[{rho},
         rho = forest2part[Sverbal[t]];
         AllTrue[rho, SSQ[subtab[t, #]] &]]
```

Do the same for a left-zero semigroup

```
In[]:= LeftZeroQ[t_] := With[{n = Length[t]}, Module[{flag = True},
          Do[If[t[[x, y]] ≠  x, flag = False; Break[]], {x, 1, n}, {y, 1, n}];
          flag]]

In[]:= lzpairs[t_] := With[{n = Length[t]},
         DeleteDuplicates[Flatten[Table[{t[[x, y]], x}, {x, 1, n}, {y, 1, n}], 1]]]

In[]:= Lzverbal[t_] := cg[lzpairs[t], ETlist[t], Length[t]]
```

Left-zero semigroup of size n

```
In[]:= lzsg[n_] := Table[x, {x, 1, n}, {y, 1, n}]
```

Malcev product

malcevprod[a,b] returns an algebra whose replica is the table a, and whose congruence classes are the members of b. Does not assume commutativity.

In[∘]:=
```
maltsevprod[a_, b_] := Module[{m, n, k, c, r, j, fi, fj, ti, tj, fk},
  m = Length /@ b;
  n = Total[m];
  k = Length[a];
  j = 0; r = ConstantArray[0, k];
  Do[r[[i]] = Range[m[[i]]] + j;
   j = m[[i]] + j,
   {i, 1, k}];
  c = ConstantArray[0, {n, n}];
  Do[{fi, ti} = First[Position[r, i]];
   {fj, tj} = First[Position[r, j]];
   fk = a[[fi, fj]];
   If[fi == fj,
    c[[i, j]] = r[[fi, b[[fi, ti, tj]]]],
    c[[i, j]] = RandomChoice[r[[fk]]]],
   {i, 1, n}, {j, 1, n}];
  c]
```

I borrowed RandomSl from S (SS).nb

In[∘]:=
```
b = RandomSl[4, 7]; Length[b]
```

Out[∘]= 7

In[∘]:=
```
RelationGraph[coverssl[b], Range[7]]
```

Out[∘]=



In[∘]:=
```
alist = lzsg /@ {2, 5, 3, 5, 4, 4, 3};
```

In[∘]:=
```
mf /@ alist
```

Out[∘]=

$$\left\{\begin{pmatrix}1&1\\2&2\end{pmatrix},\begin{pmatrix}1&1&1&1&1\\2&2&2&2&2\\3&3&3&3&3\\4&4&4&4&4\\5&5&5&5&5\end{pmatrix},\begin{pmatrix}1&1&1\\2&2&2\\3&3&3\end{pmatrix},\begin{pmatrix}1&1&1&1&1\\2&2&2&2&2\\3&3&3&3&3\\4&4&4&4&4\\5&5&5&5&5\end{pmatrix},\begin{pmatrix}1&1&1&1\\2&2&2&2\\3&3&3&3\\4&4&4&4\end{pmatrix},\begin{pmatrix}1&1&1&1\\2&2&2&2\\3&3&3&3\\4&4&4&4\end{pmatrix},\begin{pmatrix}1&1&1\\2&2&2\\3&3&3\end{pmatrix}\right\}$$

In[∘]:=
```
c = maltsevprod[b, alist]
```

In[∘]:=
```
toalpha[c_] := c /. Thread[Range[26] → Characters["abcdefghijklmnopqrstuvwxyz"]]
```

Permute the algebra

In[∘]:=
```
p = RandomSample[Range[26]]
```

Out[∘]=
```
{16, 15, 1, 8, 24, 21, 13, 14, 6, 22, 2,
 19, 5, 3, 7, 20, 18, 25, 9, 23, 17, 10, 12, 26, 4, 11}
```

In[∘]:=
```
pinv = InversePermutation[p]
```

Out[∘]=
```
{3, 11, 14, 25, 13, 9, 15, 4, 19, 22, 26,
 23, 7, 8, 2, 1, 21, 17, 12, 16, 6, 10, 20, 5, 18, 24}
```

In[∘]:=
```
d = Table[pinv[[c[[p[[x]], p[[y]]]]]], {x, 1, 26}, {y, 1, 26}];
```

In[∘]:=
```
toalpha[d] // mf
```

In[∘]:=
```
formatpartitions[]
```

Test for membership in a Maltsev product

Here is a binar of size 26

Out[∘]//MatrixForm=

a	b	c	v	g	k	g	w	o	k	c	a	o	n	i	k	a	w	v	k	a	d	w	g	i	b
h	b	k	n	h	k	b	b	i	k	c	h	o	y	n	k	g	b	n	k	w	n	b	z	m	b
k	c	c	k	k	c	k	c	k	c	c	k	c	c	k	c	k	c	c	k	c	c	k	c	k	c
v	m	k	d	i	c	o	o	n	c	k	v	y	o	o	c	d	i	d	k	v	d	n	i	o	o
w	h	k	o	e	f	g	g	m	j	k	w	o	o	o	p	z	e	m	t	b	y	g	e	y	h
c	k	k	k	j	f	c	k	c	f	k	k	k	c	c	f	k	p	c	f	k	k	c	f	k	k
z	g	k	n	h	k	g	g	o	c	k	w	y	i	i	c	w	z	i	c	b	i	g	h	y	g
w	h	c	n	z	k	h	h	i	c	c	w	i	y	o	c	w	h	i	c	h	i	h	g	n	h
o	i	k	o	o	c	i	m	i	k	c	m	i	i	i	k	m	i	m	k	y	o	i	y	i	y
k	k	k	k	p	j	c	k	c	j	c	c	c	c	c	j	k	p	k	j	k	c	c	t	k	c
k	k	k	k	c	c	k	c	k	k	k	k	c	k	c	c	c	c	c	k	c	c	k	k	k	k
l	h	c	d	g	k	b	z	y	c	c	l	y	m	y	c	l	w	v	k	l	s	w	g	n	z
y	o	c	y	m	k	o	n	m	k	c	o	m	m	m	k	y	o	m	k	y	m	m	n	m	n
m	n	c	n	o	c	y	o	n	c	k	i	n	n	n	c	o	y	y	c	m	i	o	i	n	y
y	o	k	i	i	k	m	i	o	c	c	o	o	o	o	k	o	n	m	c	m	m	o	m	o	y
c	k	c	c	j	p	c	k	c	p	k	k	c	k	c	p	c	j	c	p	k	c	k	p	k	c
q	h	k	v	z	k	h	w	m	c	k	q	n	m	m	k	q	w	v	c	q	d	z	w	y	g
z	g	k	i	r	j	b	w	y	j	k	g	i	o	m	f	w	r	m	p	h	y	g	r	m	b
v	m	k	s	o	k	n	y	y	k	k	s	n	y	o	c	d	y	s	k	d	s	y	o	y	n
k	c	k	k	j	t	c	c	c	t	c	c	k	k	k	t	k	j	k	t	k	c	k	t	c	c
u	h	c	d	g	c	z	h	y	k	k	u	o	n	o	k	u	h	d	c	u	v	b	b	n	w
d	o	k	v	n	c	y	i	n	k	c	s	m	m	m	k	v	n	v	c	d	v	m	n	i	m
b	w	c	o	g	c	w	w	o	c	k	z	m	i	o	c	g	z	y	k	b	o	w	w	i	w
z	h	c	y	x	j	z	w	n	j	c	w	m	i	i	t	z	x	i	p	w	i	g	x	m	z
n	n	k	n	o	k	n	m	y	k	c	i	y	y	y	c	i	o	i	k	i	y	o	n	y	m
z	z	k	i	g	c	z	z	m	k	k	g	y	n	n	k	w	w	y	c	h	m	z	g	o	z

Is d a member of LZ ∘ Sl?

In[∘]:=
```
Timing[toalpha[forest2part[lambda = Sverbal[d]]]]
```

Out[∘]=
```
{0.102538,
 |d, s, v|f, j, p, t|c, k|a, l, q, u|e, r, x|b, g, h, w, z|i, m, n, o, y|}
```

```
In[•]:= mf[toalpha[d[[#, #]]]] & /@ List @@ forest2part[lambda]
```

$$\text{Out}=\left\{\begin{pmatrix} d & d & d \\ s & s & s \\ v & v & v \end{pmatrix}, \begin{pmatrix} f & f & f & f \\ j & j & j & j \\ p & p & p & p \\ t & t & t & t \end{pmatrix}, \begin{pmatrix} c & c \\ k & k \end{pmatrix}, \begin{pmatrix} a & a & a & a \\ l & l & l & l \\ q & q & q & q \\ u & u & u & u \end{pmatrix}, \begin{pmatrix} e & e & e \\ r & r & r \\ x & x & x \end{pmatrix}, \begin{pmatrix} b & b & b & b & b \\ g & g & g & g & g \\ h & h & h & h & h \\ w & w & w & w & w \\ z & z & z & z & z \end{pmatrix}, \begin{pmatrix} i & i & i & i & i \\ m & m & m & m & m \\ n & n & n & n & n \\ o & o & o & o & o \\ y & y & y & y & y \end{pmatrix}\right\}$$

Is d a member of Sl ∘ LZ?

```
In[•]:= Timing[toalpha[forest2part[lambda = Lzverbal[d]]]]
Out[•]= {0.005259,
  |a, b, c, d, e, f, g, h, i, j, k, l, m, n, o, p, q, r, s, t, u, v, w, x, y, z|}
```